CDI 4 - Respostas para a Prova 4

Engenharia Elétrica, Turma A
Joinville, 17 de novembro de 2008
Prof. Fernando D. Sasse

1 *Mostre que*

$$f(z) = \frac{1 - e^{2iz}}{z^2}$$

tem um pólo simples em $z = 0$. *Determine o valor principal da integral*

$$\int_{-\infty}^{+\infty} \frac{sin^2 x}{x^2}\, dx\,.$$

Use o resultado para mostrar que

$$\int_0^{\infty} \frac{\sin^2 x}{x^2}\, dx = \frac{\pi}{2}\,.$$

> $f := \frac{1 - \exp(2 \cdot I \cdot z)}{z^2}$

$$f := \frac{1 - e^{2\,I\,z}}{z^2} \qquad (1)$$

>

Examinemos o caráter de $z = 0$ em *1/f* :

> $g := \frac{1}{f}$

$$g := \frac{z^2}{1 - e^{2\,I\,z}} \qquad (2)$$

> $limit(g, z = 0)$

$$0 \qquad (3)$$

> $dg := diff(g, z)$

$$dg := \frac{2\,I\,z^2\,e^{2\,I\,z}}{\left(1 - e^{2\,I\,z}\right)^2} + \frac{2\,z}{1 - e^{2\,I\,z}} \qquad (4)$$

> $limit(dg, z = 0)$

$$\frac{1}{2}\,I \qquad (5)$$

que já é diferente de zero. Portanto, g tem um zero de primeira ordem em $z = 0$ e f tem um pólo simples em $z = 0$. O valor principal da função é dado por

$$P\int_{-\infty}^{\infty} f\,\mathrm{d}x = \pi\, i\left(\sum \mathit{residuos\ no\ eixo\ x}\right) + 2\,\pi\, i\left(\sum \mathit{residuos\ no\ semi-\ plano\ superior}\right)$$

O resíduo de f em $z = 0$ é,

> $R := residue(f, z = 0)$

$$R := -2\,\mathrm{I} \tag{6}$$

ou seja,

$$P\int_{-\infty}^{\infty} \frac{1 - \mathrm{e}^{2\,i\,x}}{x^2}\mathrm{d}x = -2\,i^2\pi = 2\,\pi$$

Notemos agora que

$1 - \mathrm{e}^{2\,i\,x} = 1 - \cos(2\,x) - i\sin(2\,x) = 1 - \cos^2(x) - 2\,i\cos(x)\sin(x) + \sin^2(x) = 2\sin^2 x - 2\,i\cos x\sin x$

ou seja,

$$\mathrm{Re}\left(1 - \mathrm{e}^{2\,i\,x}\right) = 2\sin^2 x$$

e

$$P\int_{-\infty}^{\infty} \frac{\sin^2 x}{x^2}\,dx = \frac{1}{2}\,\Re\,P\int_{-\infty}^{\infty} \frac{\left(1 - \mathrm{e}^{2\,i\,x}\right)}{x^2}\,dx = \pi$$

Como o integrando é simétrico com relação à origem, podemos afirmar que

$$\int_{0}^{\infty} \frac{\sin^2 x}{x^2}\,dx = \frac{\pi}{2}$$

2 Seja $f(t) = |t|$, $-2 < t < 2$, de período $T = 4$.
(a) Faça o gráfico desta função.
(b) Determine ao menos 4 termos não-nulos da série de Fourier complexa de $f(t)$.
(c) Determine ao menos 5 termos não nulos da série de Fourier trigométrica de $f(t)$.

(a)

> $restart$

> $T := 4:$

> $f := x \rightarrow \mathrm{abs}(x)$

$$f := x \rightarrow |x| \tag{7}$$

> $f := x \rightarrow \mathrm{abs}(x);$
> $f1 := x \rightarrow f\left(x - \mathrm{T}\cdot\mathrm{floor}\left(\left(x + \frac{T}{2}\right)/T\right)\right);$
> $plot(f1(x), x = -2\cdot T..2\cdot T,$
> $color = COLOR(RGB, .4, 0, 1), thickness = 2);$

$$f := x \rightarrow |x|$$

$$f1 := x \to f\left(x - T\,\mathrm{floor}\left(\frac{x + \frac{1}{2}T}{T}\right)\right)$$



(b)

```
> assume(k, integer);
> L := T/2;
```

$$L := 2 \tag{8}$$

```
> c:=k->int(f(t)*exp(-I*k*Pi*t/L), t = -L .. L)/(2*L);
```

$$c := k \to \frac{1}{2}\,\frac{\int_{-L}^{L} f(t)\,e^{-\frac{I k \pi t}{L}}\,dt}{L} \tag{9}$$

```
>  c(k);
```

$$-\frac{2\,(-1)^{1+k\sim} + 2}{k\sim^2\,\pi^2} \tag{10}$$

```
> c(0);
```

$$1 \tag{11}$$

```
> FS :=h-> c(0)+sum(c(n)*exp(I*n*Pi*t/L), n = 1 .. h)+sum(c(n)*exp
(I*n*Pi*t/L), n = -h .. -1);
```

$$FS := h \to c(0) + \sum_{n=1}^{h} c(n)\,e^{\frac{I n \pi t}{L}} + \sum_{n=-h}^{-1} c(n)\,e^{\frac{I n \pi t}{L}} \tag{12}$$

```
> FS(4);
```

$$1 - \frac{4\,e^{\frac{1}{2} I \pi t}}{\pi^2} - \frac{4}{9}\,\frac{e^{\frac{3}{2} I \pi t}}{\pi^2} - \frac{4}{9}\,\frac{e^{-\frac{3}{2} I \pi t}}{\pi^2} - \frac{4\,e^{-\frac{1}{2} I \pi t}}{\pi^2} \tag{13}$$

(c) Os coeficientes a_k e b_k na correspondente série trigonométrica para $f(t)$ são dados por

```
> a:=k->2*Re(c(k)); b:=k->-2*Im(c(k));
```

$$a := k \to 2\,\Re(c(k))$$

$$b := k \to -2\,\Im(c(k)) \tag{14}$$

```
> a(k);b(k);
```

$$-\frac{2\left(2\,(-1)^{1+k\sim}+2\right)}{k\sim^2\,\pi^2}$$

$$0 \tag{15}$$

```
> FStrig := h-> c(0)+sum(a(n)*cos(n*Pi*t/L)+b(n)*sin(n*Pi*t/L), n = 1 .. h);
```

$$FStrig := h \rightarrow c(0) + \sum_{n=1}^{h}\left(a(n)\cos\left(\frac{n\,\pi\,t}{L}\right) + b(n)\sin\left(\frac{n\,\pi\,t}{L}\right)\right) \tag{16}$$

```
> FStrig(5);
```

$$1 - \frac{8\cos\left(\frac{1}{2}\,\pi\,t\right)}{\pi^2} - \frac{8}{9}\,\frac{\cos\left(\frac{3}{2}\,\pi\,t\right)}{\pi^2} - \frac{8}{25}\,\frac{\cos\left(\frac{5}{2}\,\pi\,t\right)}{\pi^2} \tag{17}$$

```
> plot([f1(t), seq(FStrig(i), i = 1 .. 5)], t = -2*T .. 2*T, t, color = [black, red, blue, green, magenta, coral, brown, navy], linestyle = [3, `$`(1, 5)]);
```




```
>
```

3 *Utilize série de Fourier para determinar a corrente de estado estacionário $I(t)$ de um circuito RLC em série, com $R = 40\,ohms$, $L = 10H$, $C := 10^{-5}F$, com impulso externo $E(t) = t$ para $-1 < t < 1$, com $E(t+2) = E(t)$. Encontre ao menos três termos da série para $I(t)$.*

Solução.

A forma do sinal de entrada é dada a seguir:

```
> restart;
> T := 2:
> E := t -> t-T*floor((t+T/2)/T);
  plot(E(t),t=-2*T..2*T,color=COLOR(RGB,.4,0,.9));
```

$$E := t \to t - T\,\mathrm{floor}\left(\frac{t + \frac{1}{2}\,T}{T}\right)$$



A equação do circuito é dada por

```
> eq:=L*diff(i(t),t$2)+R*diff(i(t),t)+i(t)/C=diff(En(t), t);
```

$$eq := L\left(\frac{d^2}{dt^2}\,i(t)\right) + R\left(\frac{d}{dt}\,i(t)\right) + \frac{i(t)}{C} = \frac{d}{dt}\,En(t) \quad \textbf{(18)}$$

```
>
```

Inicialmente desenvolvemos $E(t)$ numa série de Fourier:

```
> assume(k, integer);
> L := T/2;
```

$$L := 1 \quad \textbf{(19)}$$

```
> c:=k->int(t*exp(-I*k*Pi*t/L), t = -L .. L)/(2*L);
```

$$c := k \to \frac{1}{2}\,\frac{\int_{-L}^{L} t\,e^{-\frac{I k \pi t}{L}}\,dt}{L} \quad \textbf{(20)}$$

```
>  c(k);
```

$$\frac{I\,(-1)^{k\sim}}{k\sim\,\pi} \quad \textbf{(21)}$$

```
> c(0);
```

$$0 \quad \textbf{(22)}$$

```
> FS :=h-> c(0)+sum(c(n)*exp(I*n*Pi*x/L), n = 1 .. h)+sum(c(n)*exp
  (I*n*Pi*x/L), n = -h .. -1);
```

$$FS := h \to c(0) + \sum_{n=1}^{h} c(n)\,e^{\frac{I n \pi x}{L}} + \sum_{n=-h}^{-1} c(n)\,e^{\frac{I n \pi x}{L}} \quad \textbf{(23)}$$

Os coeficientes a_k e b_k na correspondente série trigonométrica para $f(x)$ são dados por

```
> a:=k->2*Re(c(k)); b:=k->-2*Im(c(k));
```

$$a := k \to 2\,\Re(c(k))$$

$$b := k \rightarrow -2\, \Im(c(k)) \tag{24}$$

```
> a(k);b(k);
```

$$0$$

$$-\frac{2\,(-1)^{k\sim}}{k\sim\pi} \tag{25}$$

> $ES := (t, n) \rightarrow Sum\left(b(k) * \sin\left(k \cdot \text{Pi} * \frac{t}{L}\right), k = 1..n\right);$

$$ES := (t, n) \rightarrow \sum_{k=1}^{n} b(k) \sin\left(\frac{k\,\pi\,t}{L}\right) \tag{26}$$

Os primeiros termos da série são:

> $value(ES(t, 7))$

$$\frac{2\sin(\pi t)}{\pi} - \frac{\sin(2\pi t)}{\pi} + \frac{2}{3}\frac{\sin(3\pi t)}{\pi} - \frac{1}{2}\frac{\sin(4\pi t)}{\pi} + \frac{2}{5}\frac{\sin(5\pi t)}{\pi} - \frac{1}{3}\frac{\sin(6\pi t)}{\pi} + \frac{2}{7}\frac{\sin(7\pi t)}{\pi} \tag{27}$$

A solução de estado estacionário é da forma

> $assume(n, integer)$

> $i := (t) \rightarrow A \cdot \cos\left(\frac{n \cdot \pi}{L} t\right) + B \cdot \sin\left(\frac{n \cdot \pi}{L} t\right)$

$$i := t \rightarrow A \cos\left(\frac{n\,\pi\,t}{L}\right) + B \sin\left(\frac{n\,\pi\,t}{L}\right) \tag{28}$$

> $En := t \rightarrow b_n \cdot \sin\left(\frac{n \cdot \pi}{L} \cdot t\right)$

$$En := t \rightarrow b_n \sin\left(\frac{n\,\pi\,t}{L}\right) \tag{29}$$

> eq

$$-A\cos(n\sim\pi t)\, n\sim^2 \pi^2 - B\sin(n\sim\pi t)\, n\sim^2\pi^2 + R\,(-A\sin(n\sim\pi t)\, n\sim\pi + B\cos(n\sim\pi t)\, n\sim\pi) + \frac{A\cos(n\sim\pi t) + B\sin(n\sim\pi t)}{C} = b_{n\sim}\cos(n\sim\pi t)\, n\sim\pi \tag{30}$$

> $eq1 := expand(lhs(eq) - rhs(eq))$

$$eq1 := -A\cos(n\sim\pi t)\, n\sim^2\pi^2 - B\sin(n\sim\pi t)\, n\sim^2\pi^2 - R\,A\sin(n\sim\pi t)\, n\sim\pi + R\,B\cos(n\sim\pi t)\, n\sim\pi + \frac{A\cos(n\sim\pi t)}{C} + \frac{B\sin(n\sim\pi t)}{C} - b_{n\sim}\cos(n\sim\pi t)\, n\sim\pi \tag{31}$$

> $eq2 := collect(eq1, \cos(n \cdot t))$

$$eq2 := -A\cos(n\sim\pi t)\, n\sim^2\pi^2 - B\sin(n\sim\pi t)\, n\sim^2\pi^2 - R\,A\sin(n\sim\pi t)\, n\sim\pi \tag{32}$$

$$+ R B \cos(n\sim \pi t)\, n\sim \pi + \frac{A \cos(n\sim \pi t)}{C} + \frac{B \sin(n\sim \pi t)}{C} - b_{n\sim} \cos(n\sim \pi t)\, n\sim \pi$$

> $eq3 := collect\left(eq2, \sin\left(\frac{n \cdot \mathrm{Pi}}{L} \cdot t\right)\right)$

$$eq3 := \left(-B\, n\sim^2 \pi^2 + \frac{B}{C} - R A\, n\sim \pi\right) \sin(n\sim \pi t) - A \cos(n\sim \pi t)\, n\sim^2 \pi^2 - b_{n\sim} \cos(n\sim \pi t)\, n\sim \pi + R B \cos(n\sim \pi t)\, n\sim \pi + \frac{A \cos(n\sim \pi t)}{C} \quad \textbf{(33)}$$

> $ee1 := coeff\left(eq3, \sin\left(\frac{n \cdot \pi}{L} \cdot t\right)\right) = 0$

$$ee1 := -B\, n\sim^2 \pi^2 + \frac{B}{C} - R A\, n\sim \pi = 0 \quad \textbf{(34)}$$

> $ee2 := coeff\left(eq3, \cos\left(\frac{n \cdot \pi}{L} \cdot t\right)\right) = 0$

$$ee2 := -A\, n\sim^2 \pi^2 - b_{n\sim}\, n\sim \pi + R B\, n\sim \pi + \frac{A}{C} = 0 \quad \textbf{(35)}$$

> $sol := solve(\{ee1, ee2\}, \{A, B\})$

$$sol := \left\{A = -\frac{b_{n\sim}\, n\sim \pi C \left(n\sim^2 \pi^2 C - 1\right)}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1}, B = \frac{b_{n\sim}\, n\sim^2 \pi^2 C^2 R}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1}\right\} \quad \textbf{(36)}$$

> $assign(sol)$

> $A := subs\left(b_n = b(n), A\right)$

$$A := \frac{2\, (-1)^{n\sim} C \left(n\sim^2 \pi^2 C - 1\right)}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1} \quad \textbf{(37)}$$

> $B := subs\left(b_n = b(n), B\right)$

$$B := -\frac{2\, (-1)^{n\sim} n\sim \pi C^2 R}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1} \quad \textbf{(38)}$$

> $A := unapply(A, n)$

$$A := n\sim \rightarrow \frac{2\, (-1)^{n\sim} C \left(n\sim^2 \pi^2 C - 1\right)}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1} \quad \textbf{(39)}$$

> $B := unapply(B, n)$

$$B := n\sim \rightarrow -\frac{2\, (-1)^{n\sim} n\sim \pi C^2 R}{n\sim^4 \pi^4 C^2 - 2\, n\sim^2 \pi^2 C + R^2 n\sim^2 \pi^2 C^2 + 1} \quad \textbf{(40)}$$

>

As soluções são da forma:

> $i := Sum\left(A(j)\cdot\cos\left(\frac{j\cdot\pi\cdot n}{L}t\right) + B(j)\cdot\sin\left(\frac{j\cdot n\cdot\pi}{L}t\right), j=1..n\right)$

$$i := \sum_{j=1}^{n\sim}\left(\frac{2\,(-1)^j\,C\,(j^2\,\pi^2\,C-1)\,\cos(j\,\pi\,n\sim t)}{j^4\,\pi^4\,C^2-2\,j^2\,\pi^2\,C+R^2\,j^2\,\pi^2\,C^2+1}-\frac{2\,(-1)^j\,j\,\pi\,C^2\,R\sin(j\,\pi\,n\sim t)}{j^4\,\pi^4\,C^2-2\,j^2\,\pi^2\,C+R^2\,j^2\,\pi^2\,C^2+1}\right) \quad \textbf{(41)}$$

Utilizando três termos e os valores de C, L e R dados no problema temos

> $n := 3;\ C := 10^{-5};\ L := 10;\ R := 40;$

$$n := 3$$

$$C := \frac{1}{100000}$$

$$L := 10$$

$$R := 40 \quad \textbf{(42)}$$

> $i := Sum\left(A(j)\cdot\cos\left(\frac{j\cdot\pi\cdot n}{L}t\right) + B(j)\cdot\sin\left(\frac{j\cdot n\cdot\pi}{L}t\right), j=1..n\right)$

$$i := \sum_{j=1}^{3}\left(\frac{1}{50000}\,\frac{(-1)^j\left(\frac{1}{100000}\,j^2\,\pi^2-1\right)\cos\left(\frac{3}{10}\,j\,\pi\,t\right)}{\frac{1}{10000000000}\,j^4\,\pi^4-\frac{31}{1562500}\,j^2\,\pi^2+1}-\frac{1}{125000000}\,\frac{(-1)^j\,j\,\pi\sin\left(\frac{3}{10}\,j\,\pi\,t\right)}{\frac{1}{10000000000}\,j^4\,\pi^4-\frac{31}{1562500}\,j^2\,\pi^2+1}\right) \quad \textbf{(43)}$$

>

> $evalf(\%)$

$$0.00002000194252\cos(0.9424777961\,t)+2.513766326\;10^{-8}\sin(0.9424777961\,t)-0.00002000777232\cos(1.884955592\,t)-5.030487600\;10^{-8}\sin(1.884955592\,t)+0.00002001749608\cos(2.827433388\,t)+7.553127411\;10^{-8}\sin(2.827433388\,t) \quad \textbf{(44)}$$

> $plot(i, t=0..7\cdot T)$



>

4 *Determine graficamente o espectro de Fourier de um pulso quadrado de altura 1 e largura a. Interprete o resultado.*

```
> restart
```

Gráfico do pulso com $a = 1$.

```
> f := t -> if abs(t)<1 then 1 else 0 end if:
  plot('f(t)',t=-3..3,0..1.2,color=red,thickness=2,ytickmarks=3);
  ;
```



A integral de Fourier é dada por

$$F(\omega) = \int_{-\infty}^{\infty} f(t)\, e^{-i\omega t}\, dx = \int_{-a}^{a} e^{-i\omega t}\, dt = \left.\frac{e^{-i\omega t}}{-i\omega}\right|_{-a}^{a} = \frac{e^{-ia\omega} - e^{ia\omega}}{-i\omega}$$

$$= \frac{2\left(\frac{e^{ia\omega} - e^{-ia\omega}}{2i}\right)}{\omega}$$

$$= \frac{2\sin a\omega}{\omega}$$

No Maple,

```
> int(exp(-I*omega*t), t = -a .. a);
```

$$FS := -\frac{I\left(e^{I\omega a} - e^{-I\omega a}\right)}{\omega} \quad (45)$$

```
> FS := convert(%, sin);
```

$$\frac{2\sin(\omega a)}{\omega} \quad (46)$$

```
> plot(2*sin(w)/w, w = -10 .. 10, color = COLOR(RGB, .4, 0, .9),
  title = `Transformada de Fourier F(w) of f(t)`);
```



Neste gráfico o eixo horizontal corresponde às frequências dos harmônicos (infinitos) necessários para formar o pulso. O eixo vertical corresponde às amplitudes.

5 *O resultado da amostragem de um sinal nos tempos* $[0, 1/4, 1/2, 3/4]$ *é* $[3, 5, -3, -5]$. *Aplique a transformada de Fourier discreta para encontrar amplitudes e freqüências das componentes* sin *e* cos *deste sinal.*

```
> with(LinearAlgebra):
```

> $X := \langle 3, 5, -3, -5 \rangle$

$$X := \begin{bmatrix} 3 \\ 5 \\ -3 \\ -5 \end{bmatrix} \tag{47}$$

> $n := 4:$

> $F := (i, j) \rightarrow \exp\left(-\frac{2 \cdot \pi \cdot I}{n} \cdot (i-1) \cdot (j-1)\right);$

$$F := (i, j) \rightarrow \mathrm{e}^{-\frac{2 I \pi (i-1)(j-1)}{n}} \tag{48}$$

> $FF := Matrix(4, F)$

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & -I & -1 & I \\ 1 & -1 & 1 & -1 \\ 1 & I & -1 & -I \end{bmatrix} \tag{49}$$

> $F4 := \%:$

> $Y := \frac{2}{n} \cdot F4.X$

$$Y := \begin{bmatrix} 0 \\ 3 - 5\,\mathrm{I} \\ 0 \\ 3 + 5\,\mathrm{I} \end{bmatrix} \tag{50}$$

Ou seja,

$$Y = \begin{bmatrix} 0 \\ 3 \\ 0 \\ 3 \end{bmatrix} + i \begin{bmatrix} 0 \\ -5 \\ 0 \\ 5 \end{bmatrix} = 3 \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix} + 3 \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} + 5\,i \begin{bmatrix} 0 \\ -1 \\ 0 \\ 0 \end{bmatrix} + 5\,i \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} = 3\,(e_1 + e_3) + 5\,i(-e_1 + e_3)$$

A parte real de Y nos diz que há uma componente de cosseno com amplitude 3 e frequência 1, enquanto que a parte imaginária nos diz que há uma componente de seno com amplitude 5 e frequência 1 (somente a informação na primeira metade de Y é necessária). O sinal no domínio do tempo é, portanto, $x(t) = 3\cos(2\,\pi\,t) + 5\sin(2\,\pi\,t)$.